

```

10 REM FILENAME: SLIDELBL.BAS
20 REM Original program written by Wally Hampton (c)1987
30 REM and it appeared in OUTDOOR PHOTOGRAPHER magazine
40 REM in the June 1987 issue.
50 REM Additional bells, whistles & bugs have been added
60 REM by Michael F. Havelin (c)1987.
70 REM This program is for _FREE_ distribution. It may not be sold.
80 REM
90 CLS
100 REM
110 REM NEXT STATEMENTS SET AN EPSON MX-80 PRINTER
120 REM SO IT WILL PRINT AT 8 LINES PER INCH AND IN
130 REM COMPRESSED PRINT
140 REM
150 LPRINT CHR$(27);CHR$(15)
160 LPRINT CHR$(27);CHR$(48)
170 REM S1$="          "
180 S1$= "                "
190 GOTO 270
200 L1$= "THIS IS MAX LENGTH OF PRITING"
210 FOR CL = 1 TO 3
220     LPRINT S1$;L1$;CHR$(10)
230     LPRINT S1$;L1$;CHR$(10)
240     LPRINT S1$;L1$;CHR$(10)
250     LPRINT;CHR$(10)
260     NEXT CL
270 LPRINT
280 INPUT "IS PRINTER LINED-UP PROPERLY? (Y/N)";L2$
290 IF L2$ = "N" GOTO 200
300 IF L2$ = "n" GOTO 200
310 L1$= " "
320 L2$= " "
330 L3$= " "
340 NL = 0
350 REM SELECT TO GENERATE SLIDE NUMBERS OR NOT
360 PRINT:PRINT
370 INPUT "DO YOU WANT TO GENERATE CONSECUTIVE SLIDE NUMBERS";YN$
380 IF YN$ = "Y" GOTO 620
390 IF YN$ = "y" GOTO 620
400 PRINT:PRINT
410 PRINT "TYPE LABEL TEXT..."
420 PRINT "LIMIT TEXT LENGTH TO DOTTED LINE"
430 PRINT "-----"
440 LINE INPUT L1$
450 IF LEN(L1$) < 1 GOTO 580
460 LINE INPUT L2$
470 LINE INPUT L3$
480 PRINT "-----"
490 INPUT "# OF LABELS";NL
500 PRINT:PINT
510 IF NL < 1 GOTO 580
520 FOR CL = 1 TO NL
530     LPRINT S1$;L1$;CHR$(10)
540     LPRINT S1$;L2$;CHR$(10)
550     LPRINT S1$;L3$;CHR$(10)
560     LPRINT;CHR$(10)
570     NEXT CL

```

```

580 INPUT "PRINT MORE LABELS? (Y/N) ";L4$
590 IF L4$ = "N" GOTO 1010
600 IF L4$ = "n" GOTO 1010
610 GOTO 310
620 PRINT:PRINT
630 INPUT "NEXT SLIDE NUMBER ==> ";C1
640 PRINT
650 NL = 0
660 PRINT
670 PRINT "TYPE 3 LINES OF LABEL TEXT"
680 PRINT "LIMIT TEXT LENGTH TO DOTTED LINE"
690 PRINT "END 1st LINE AT STAR =>"
700 PRINT "          * "
710 PRINT "-----"
720 LINE INPUT L1$
730 IF LEN(L1$) < 1 GOTO 880
740 LINE INPUT L2$
750 LINE INPUT L3$
760 PRINT "-----"
770 PRINT:PRINT
780 INPUT "# OF LABELS";NL
790 IF NL < 1 GOTO 880
800 FOR CL = 1 TO NL
810     LPRINT S1$;"#";C1;L1$;CHR$(10)
820     LPRINT S1$;L2$;CHR$(10)
830     LPRINT S1$;L3$;CHR$(10)
840     LPRINT;CHR$(10)
850     C1 = C1 + 1
860     NEXT CL
870 PRINT:PRINT
880 INPUT "PRINT MORE LABELS? (Y/N) ";L4$
890 PRINT:PRINT
900 IF L4$ = "N" GOTO 930
910 IF L4$ = "n" GOTO 930
920 GOTO 660
930 PRINT:PRINT
940 PRINT "NEXT STARTING SLIDE NUMBER WILL BE #"; C1
950 PRINT:PRINT
960 PRINT "WRITE IT DOWN!!!!!!!"
970 PRINT
980 FOR CL = 1 TO 4000
990 NEXT CL
1000 IF CL < 4000 GOTO 990
1010 STOP

```